miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University
WEIXI MA, Indiana University
DANIEL P. FRIEDMAN, Indiana University

The syntax of a programming language should reflect its semantics. When writing a **cond**^{e} expression in miniKanren, a programmer would expect all clauses share the same chance of being explored, as these clauses are written in parallel. The existing search strategy, interleaving DFS (DFS$_\mathrm{i}$), however, prioritize its clauses by the order how they are written down. Similarly, when a **cond**^{e} is followed by another goal conjunctively, a programmer would expect answers in parallel share the same chance of being explored. Again, the answers by DFS$_\mathrm{i}$ is different from the expectation. We have devised three new search strategies that have different level of fairness in `disj` and `conj`.

ACM Reference Format:
Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. miniKanren with fair search strategies. 1, 1 (May 2019), 12 pages. https://doi.org/10.1145/nnnnnnn.nnnnnnn

1 INTRODUCTION

miniKanren is a family of relational programming languages. miniKanren programs, especially relational interpreters, have been proven to be useful in solving many problems by Byrd et al. [1]. A subtlety in writing miniKanren programs with large **cond**^{e} expressions, such as relational interpreters, is that the order of **cond**^{e} clauses sometimes affect the speed considerably. This phenomenon also appears when running miniKanren programs with the implementation in Friedman et al. [2], one of the most well-understood implementation. This is because the search strategy of this implementation, interleaving depth-first search (DFS$_\mathrm{i}$), gives left clauses higher "search priority" when a **cond**^{e} has more than two clauses. The situation is exacerbated as the number of clauses increases. This unfair treatment causes two problems when a **cond**^{e} expression has many clauses: the right-most clause can hardly contribute to query result; and programmers might need to tweak the clause order to maximize their programs' efficiency, depending on the distribution of queries.

The specification that every clause in the same **cond**^{e} is given equal "search priority" is called fair `disj`. And search strategies with *almost-fair* `disj` should give every clauss in the same **cond**^{e} similar priority. A related concept, fair `conj` is more complicated. We defer it to the next section.

To summarize our contribution, we

- analyze an existing concept, fair `disj` by Seres et al. [6].
- propose a new concept, almost-fair `disj`.
- propose a new definition of fair `conj`.

Authors' addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

XXXX-XXXX/2019/5-ART $15.00
https://doi.org/10.1145/nnnnnnn.nnnnnnn

- propose and implement balanced interleaving depth-first search (DFS_{bi}), a new search strategy with almost-fair `disj`.
- propose and implement fair depth-first search (DFS_{f}), a new search strategy with fair `disj`.
- implement in a new way breath-first search (BFS), a search strategy with fair `disj` and fair `conj` (our code runs faster in all benchmarks and is simpler)
- prove our BFS implementation is equivalent with the one by Seres et al. [6].

2 FAIRNESS

Before going further into fairness, we would like to give a short review about state, search space, and goal, because fairness is defined in terms of them.

Every answer given by miniKanren corresponds to a state. Each state records a way how some relations can be satisfied. In our case, the state is just a subsitution associating logic variables with terms. In more expressive miniKanrens (e.g. miniKanren with with symbolic constraints [3]), states can include more information. A search space is a collection of states. And a goal is a function from a state to a search space. Goals check their inputs and produces possibly extended states. A goal might fail for some input, in which case the output search space would be empty.

Our definition of *fair* `disj` is slightly generalize from the one by Seres et al. [6]. Their definition is for binary disjunct. We generalize it to multi-arity one.

Definition 2.1 (fair `disj`). *A* `disj` *is fair iff it allocates computational resource evenly to search spaces produced by goals in the same disjunct (e.g. clauses in the same* **cond**^{e}*).*

Definition 2.2 (almost-fair `disj`). *A* `disj` *is almost-fair iff it allocates computational resource so evenly to search spaces produced by goals in the same disjunct that the maximal ratio of resources is bounded by a constant.*

The `disj` in DFS_{i} is neither fair nor almost-fair. Fair depth-first search (DFS_{f}), a new search strategy in this paper, has fair `disj`. And balanced interleaving depth-first search (DFS_{bi}), another new strategy, has almost-fair `disj`.

Breath-first search (BFS), a known strategy Seres et al. [6], has fair `disj` and *fair* `conj`.

Definition 2.3 (fair `conj`). *A* `conj` *is fair iff it allocates computational resource evenly to search spaces derived from states in the same bag, where bags are finite list of states.*

Now we elaborate fairness by running queries about repeat^o, a relational definition that relates a term `x` with a non-empty list whose elements are `x` (Fig. 1).

2.1 fair `disj`

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer. DFS_{i}, the current search strategy, however, results in many more lists of `a`s than lists of other letters. And some letters, e.g. `c` and `d`, are rarely seen. The situation would be exacerbated if **cond**^{e} contains more clauses.

```
;; iDFS (unfair disj)
> (run 12 q
    (conde
      [(repeato 'a q)]
      [(repeato 'b q)]
      [(repeato 'c q)]
```

```
> (defrel (repeat^o x out)
    (cond^e
      [(≡ `(,x) out)]
      [(fresh (res)
         (≡ `(,x . ,res) out)
         (repeat^o x res))]))
> (run 4 q
    (repeat^o '* q))
'((*) (* *) (* * *) (* * * *))
```

Fig. 1. repeat^o and an example run

```
      [(repeat^o 'd q)]))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```


We borrow the definition of *fair* `disj` from Seres et al. [6]: search strategies with *fair* `disj` should allocate resources evenly among disjunctive goals. Running the same program with $\mathrm{DFS_f}$ and BFS give the following result.

```
;; fDFS and BFS (fair disj)
> (run 12 q
    (cond^e
      [(repeat^o 'a q)]
      [(repeat^o 'b q)]
      [(repeat^o 'c q)]
      [(repeat^o 'd q)]))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

Now we are in a middle place between fair and unfair – search strategies with *almost-fair* `disj` should allocate resources so evenly among disjunctive goals that the maximal ratio of resources is bounded by a constant. Our new search strategy, $\mathrm{DFS_{bi}}$, has almost-fair `disj`. It is fair when the number of goals is a power of 2, otherwise, some goals are allocated twice as many resources than the others. In the previous example, $\mathrm{DFS_{bi}}$ gives the same result. And in the following example, where the **cond**^{e} has 5 clause, the clauses of `b`, `c`, and `d` are allocated more resources.

```
;; biDFS (almost-fair disj)
> (run 16 q
    (cond^e
      [(repeat^o 'a q)]
```

```
        [(repeat^o 'b q)]
        [(repeat^o 'c q)]
        [(repeat^o 'd q)]
        [(repeat^o 'e q)]))
'((b) (c) (d) (a)
  (b b) (c c) (d d) (e)
  (b b b) (c c c) (d d d) (a a)
  (b b b b) (c c c c) (d d d d) (e e))
```

2.2 fair `conj`

In the following program, the three **cond**^{e} clauses differ in a trivial way. So we expect lists of each letter constitute 1/4 of the answer list. Search strategies with unfair `conj` (e.g. DFS_i, DFS_f), however, give us many more lists of `a`s than lists of other letters. And some letters (e.g. lists of `c`) are rarely found. Although DFS_i's `disj` is unfair in general, it is fair when there is no call to relational definition in sub-goals, including this case. The situation would be worse if we add more **cond**^{e} clauses. The result with DFS_bi, whose `conj` is also unfair, is similar, but due to its different `disj`, the position of `b` and `c` are swapped.


```
;; iDFS
> (run 12 q
    (fresh (x)
      (cond^e
        ((≡ 'a x))
        ((≡ 'b x))
        ((≡ 'c x))
        ((≡ 'd x)))
      (repeat^o x q)))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

```
;; fDFS
> (run 12 q
    (fresh (x)
      (cond^e
        ((≡ 'a x))
        ((≡ 'b x))
        ((≡ 'c x))
        ((≡ 'd x)))
      (repeat^o x q)))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
```

```
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

Intuitively, search strategies with fair `conj` should produce each letter of lists equally frequently. Indeed, BFS does so.

```
;; BFS (fair conj)
> (run 12 q
    (fresh (x)
      (conde
        [(≡ 'a x)]
        [(≡ 'b x)]
        [(≡ 'c x)]
        [(≡ 'd x)])
      (repeato x q)))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```


A more interesting situation is when the first conjunctive goal produces infinite many answers. Consider the following example, a naive specification of fair `conj` might require search strategies to produce all sorts of singleton lists, but no longer ones, which makes the strategies *incomplete*.

c1

c2

```
;; naively fair conj
> (run 6 q
    (fresh (xs)
      (conde
        [(repeato 'a xs)]
        [(repeato 'b xs)])
      (repeato xs q)))
'(((a)) ((b))
  ((a a)) ((b b))
  ((a a a)) ((b b b)))
```

Our solution requires a search strategy with *fair* `conj` to package answers in bags, where each bag contains finite answers, and to allocate resources evenly among search spaces derived from answers in the same bag. The way to package depends on search strategy. And how to allocate resources among search

[c1] *MVC*: incomplete w.r.t. what? where's the definition of incomplete?

[c2] *LKC*: I am a bit confused. I assume completeness is a well-known concept in the context of logic programming. For example, this paper doesn't cite any source when it talks about completeness.

Hemann, Jason, et al. "A small embedding of logic programming with a simple complete search." ACM SIGPLAN Notices. Vol. 52. No. 2. ACM, 2016.

space related to different bags is unspecified. Our definition of fair `conj` is orthogonal with completeness. For example, a naively fair strategy is fair but not complete, while BFS is fair and complete. [c3]

BFS packages answers by their costs. The *cost* of a answer is its depth in the search tree (i.e. the number of calls to relational definitions required to find them) Seres et al. [6]. In the following example, every answer is a list of list of symbol. The cost of each of them is equal to the length of the inner lists plus the length of the outer list. In addition to being fair, BFS also produces answers in increasing order of cost. [c1] [c2]

```
;; BFS (fair conj)
> (run 12 q
    (fresh (xs)
      (conde
        [(repeato 'a xs)]
        [(repeato 'b xs)])
      (repeato xs q)))
'(((a)) ((b))
  ((a) (a)) ((b) (b))
  ((a a)) ((b b))
  ((a) (a) (a)) ((b) (b) (b))
  ((a a) (a a)) ((b b) (b b))
  ((a a a)) ((b b b)))
```


3 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS (DFS_{bi}) has almost-fair **disj** and unfair `conj`. The implementation of DFS_{bi} differs from DFS_{i} in the **disj** macro. We list the new **disj** with its helpers in Fig. 2. The first helper function, `split`, takes a list of goals `ls` and a procedure `k`, partitions `ls` into two sub-lists of roughly equal length, and returns the application of `k` to the two sub-lists. `disj*` takes a non-empty list of goals `gs` and returns a goal. With the help of `split`, it essentially constructs a *balanced* binary tree where leaves are elements of `gs` and nodes are `disj2`s, hence the name of this search strategy. In contrast, the **disj** in DFS_{i} constructs the binary tree with the same nodes but in the unbalanced form.

4 FAIR DEPTH-FIRST SEARCH

Fair DFS (DFS_{f}) has fair **disj** and unfair `conj`. The implementation of DFS_{f} differs from DFS_{i}'s in `disj2` (Fig. 3). `disj2` is changed to call a new and fair version of append^{∞}. $\text{append}^{\infty}_{f}$ immediately calls its helper, $\text{append}^{\infty\,\hat{}}_{f}$, with the first argument, `s?`, set to `#t`, which indicates that s^{∞} and t^{∞} haven't been swapped. The swapping happens at the third **cond** clause in the helper, where `s?` is updated accordingly. The first two **cond** clauses essentially copy the `car`s and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned above, is just for swapping. When the fourth and last clause runs, we know that both s^{∞} and t^{∞} are ended with a thunk. In this case, a new thunk is constructed. The new thunk calls the driver recursively. Here changing the order of t^{∞} and s^{∞} won't

[c3] *MVC*: Also here, complete w.r.t what?
[c1] *MVC*: Here inner and outer are very confusing. Can you be more specified?
[c2] *LKC*: updated

```
#| [Goal] x ([Goal] x [Goal] → Goal) → Goal |#
(define (split ls k)
  (cond
    [(null? ls) (k '() '())]
    [else (split (cdr ls)
                 (lambda (l1 l2)
                   (k (cons (car ls) l2) l1)))]))

#| [Goal] → Goal |#
(define (disj* gs)
  (cond
    [(null? (cdr gs)) (car gs)]
    [else
     (split gs
       (lambda (gs₁ gs₂)
         (disj2 (disj* gs₁)
                (disj* gs₂))))]))

(define-syntax disj
  (syntax-rules ()
    [(disj) fail]
    [(disj g ...) (disj* (list g ...))]))
```

Fig. 2. DFS_{bi} implementation

```
#| Goal × Goal → Goal |#
(define (disj2 g₁ g₂)
  (lambda (s)
    (append∞f (g₁ s) (g₂ s))))

#| Space × Space → Space |#
(define (append∞f s∞ t∞)
  (append∞f^ #t s∞ t∞))

#| Bool × Space × Space → Space |#
(define (append∞f^ s? s∞ t∞)
  (cond
    ((pair? s∞)
     (cons (car s∞)
       (append∞f^ s? (cdr s∞) t∞)))
    ((null? s∞) t∞)
    (s? (append∞f^ #f t∞ s∞))
    (else (lambda ()
            (append∞f (t∞) (s∞))))))
```

Fig. 3. DFS_{f} implementation

hurt the fairness (though it will change the order of answers). We swapped them back so that answers are produced in a more natural order.

5 BREADTH-FIRST SEARCH

BFS is fair in both **disj** and `conj`. Our implementation is based on $\mathrm{DFS_f}$ (not $\mathrm{DFS_i}$). All we have to do is apply two trivial changes to append-map$^\infty$. First, rename it to append-map$^\infty_f$. Second, replace its use of append$^\infty$ to append$^\infty_f$.

The implementation can be improved in two ways. First, as mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. In this implementation, however, it is unclear where this information is recorded. Second, append$^\infty_f$ is extravagant in memory usage. It makes $O(n+m)$ new `cons` cells every time, where n and m are the "length"s of input search spaces. We address these issues in the first subsection.

Both our BFS and Seres's BFS Seres et al. [6] produce answers in increasing order of cost. So it is interesting to see if they are equivalent. We prove so in Coq. The details are in the second subsection.

5.1 optimized BFS

c1 c2 c3

As mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. The cost information is recorded subtly – the `cars` of a search space have cost 0 (i.e. they are in the same bag), and the costs of answers in thunk are computed recursively then increased by one. It is even more subtle that append$^\infty_f$ and the append-map$^\infty_f$ respects the cost information. We make these facts more obvious by changing the type of search space, modifying related function definitions, and introducing a few more functions.

The new type is a pair whose `car` is a list of answers (the bag), and whose `cdr` is either a `#f` or a thunk returning a search space. A falsy `cdr` means the search space is obviously finite.

Functions related to the pure subset are listed in Fig. 4 (the others in Fig. 5). They are compared with Seres et al.'s implementation later. The first three functions in Fig. 4 are search space constructors. `none` makes an empty search space; `unit` makes a space from one answer; and `step` makes a space from a thunk. The remaining functions do the same thing as before.

Luckily, the change in append$^\infty_f$ also fixes the miserable space extravagance – the use of `append` helps us to reuse the first bag of t^∞.

Kiselyov et al. [4] has shown that a *MonadPlus* hides in implementations of logic programming system. Our BFS implementation is not an exception: `none`, `unit`, append-map$^\infty$, and append$^\infty$ correspond to `mzero`, `unit`, `bind`, and `mplus` respectively.

Functions implementing impure features are in Fig. 5. The first function, `elim`, takes a space s^∞ and two continuations k_s and k_f. When s^∞ contains some answers, k_s is called with the first answer and the rest space. Otherwise, k_f is called with no argument. Here 's' and 'f' means 'succeed' and 'fail' respectively. This function is an eliminator of search space, hence the name. The remaining functions do the same thing as before.

[c1] *MVC*: Though bag is well known, people rarely say "bagging". How about putting information in a bag, or something better?

[c2] *MVC*: What is the bagging information?

[c3] *LKC*: It's just cost... You're right. I should have be more direct.

```
(define (none)   `(()   . #f))
(define (unit s) `((,s) . #f))
(define (step f) `(()   . ,f))

(define (append∞_f s∞ t∞)
  (cons (append (car s∞) (car t∞))
    (let ([t1 (cdr s∞)]
          [t2 (cdr t∞)])
      (cond
        [(not t1) t2]
        [(not t2) t1]
        [else (lambda () (append∞_f (t1) (t2)))]))))

(define (append-map∞_f g s∞)
  (foldr
    (lambda (s t∞)
      (append∞_f (g s) t∞))
    (let ([f (cdr s∞)])
      (step (and f (lambda () (append-map∞_f g (f))))))
    (car s∞)))

(define (take-inf n s∞)
  (let loop ([n n]
             [vs (car s∞)])
    (cond
      ((and n (zero? n)) '())
      ((pair? vs)
       (cons (car vs)
         (loop (and n (sub1 n)) (cdr vs))))
      (else
       (let ([f (cdr s∞)])
         (if f (take-inf n (f)) '()))))))
```


Fig. 4. new and changed functions in optimized BFS that implements pure features

5.2 comparison with the BFS of Seres et al. [6]

In this section, we compare the pure subset of our optimized BFS with the BFS found in Seres et al. [6]. We focus on the pure subset because Silvija's system is pure. Their system represents search spaces with streams of lists of answers, where each list is a bag.

To compare efficiency, we translate her Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity in both logic programming systems and search space representations. The translated code is longer and slower. Details about difference in efficiency are in section 6.

We prove in Coq that the two BFSs are equivalent, i.e. (**run** n g) produces the same result (See supplements for the formal proof).

```
(define (elim s∞ k_s k_f)
  (let ([ss (car s∞)]
        [f (cdr s∞)])
    (cond
      [(and (null? ss) f)
       (step (lambda () (elim (f) k_s k_f)))]
      [(null? ss) (k_f)]
      [else (k_s (car ss) (cons (cdr ss) f))])))

(define (ifte g1 g2 g3)
  (lambda (s)
    (elim (g1 s)
      (lambda (s0 s∞)
        (append-map∞_f g2
          (append∞_f (unit s0) s∞)))
      (lambda () (g3 s)))))

(define (once g)
  (lambda (s)
    (elim (g s)
      (lambda (s0 s∞) (unit s0))
      (lambda () (none)))))
```

Fig. 5. new and changed functions in optimized BFS that implements impure features

6 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of search strategies. A concise description is in Table 1. A hyphen means running out of memory. The first two benchmarks are taken from Friedman et al. [2]. `reverso` is from Rozplokhas and Boulytchev [5]. Next two benchmarks about quine are modified from a similar test case in Byrd et al. [1]. The modifications are made to circumvent the need for symbolic constraints (e.g. $\neq$, `absento`). Our version generates de Bruijnized expressions and prevent closures getting into list. The two benchmarks differ in the **cond**^{e} clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to '(`I love you`). This benchmark is also inspired by Byrd et al. [1]. Again, the sibling benchmarks differ in the **cond**^{e} clause order of their relational interpreters. The first one has elimination rules (i.e. application, `car`, and `cdr`) at the end, while the other has them at the beginning. We conjecture that DFS_i would perform badly in the second case because elimination rules complicate the problem when running backward. The evaluation supports our conjecture.

In general, only DFS_i and DFS_bi constantly perform well. DFS_f is just as efficient in all benchmarks but very-recursive^o. Both BFS have obvious overhead in many cases. Among the three variants of DFS (they all have unfair `conj`), DFS_f is most resistant to clause permutation, followd by DFS_bi then DFS_i. Among the two implementation of BFS, ours constantly performs as well or better. Interestingly, every strategies with fair **disj** suffers in very-recursive^o and DFS_f performs well elsewhere. Therefore, this benchmark might be a special case. Fair `conj` imposes overhead constantly except in append^o. The reason might be that strategies with fair `conj` tend to keep more intermediate answers in the memory.

benchmark	size	DFS_i	DFS_{bi}	DFS_f	optimized BFS	Silvija's BFS
very-recursiveo	100000	579	793	2131	1438	3617
	200000	1283	1610	3602	2803	4212
	300000	2160	2836	-	6137	-
appendo	100	31	41	42	31	68
	200	224	222	221	226	218
	300	617	634	593	631	622
reverso	10	5	3	3	38	85
	20	107	98	51	4862	5844
	30	446	442	485	123288	132159
quine-1	1	71	44	69	-	-
	2	127	142	95	-	-
	3	114	114	93	-	-
quine-2	1	147	112	56	-	-
	2	161	123	101	-	-
	3	289	189	104	-	-
'(I love you)-1	99	56	15	22	74	165
	198	53	72	55	47	74
	297	72	90	44	181	365
'(I love you)-2	99	242	61	16	66	99
	198	445	110	60	42	64
	297	476	146	49	186	322

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

7 RELATED WORKS

Edward points out a disjunct complex would be 'fair' if it is a full and balanced tree Yang [7].

Silvija et al Seres et al. [6] also describe a breadth-first search strategy. We proof their BFS is equivalent to ours. But our code looks simpler and performs better in comparison with a straightforward translation of their Haskell code.

8 CONCLUSION

We analysis the definitions of fair **disj** and fair `conj`, then propose a new definition of fair `conj`. Our definition is orthogonal with completeness.

We devise three new search strategies: balanced interleaving DFS (DFS_{bi}), fair DFS (DFS_f), and BFS. DFS_{bi} has almost-fair **disj** and unfair `conj`. DFS_f has fair **disj** and unfair `conj`. BFS has both fair **disj** and fair `conj`.

Our quantitative evaluation shows that DFS_{bi} and DFS_f are competitive alternatives to DFS_i, the current search strategy, and that BFS is less practical.

We prove our BFS is equivalent to the BFS in Seres et al. [6]. Our code is shorter and runs faster than a direct translation of their Haskell code.

ACKNOWLEDGMENTS

REFERENCES

[1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).

[2] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition.*

[3] Jason Hemann and Daniel P Friedman. 2017. A framework for extending microkanren with constraints. *arXiv preprint arXiv:1701.00633* (2017).

[4] Oleg Kiselyov, Chung-chieh Shan, Daniel P Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.

[5] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming.* ACM, 18.

[6] Silvija Seres, J Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.

[7] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue 15* (2010), 11.